
1. はじめに 

  テーマは、「計数データの解析の進化」です。反応が 2 値の実験データの解析に 2x2

の分割表の解析、アミテージの傾向性検定、層がある場合 kx2x2 表に対するマンテル・

ヘンセルの検定などが多用されてきました。高橋セミナーでも第 6 回で「計数値，比率

データの解析」、第 15 回で「比の統計学入門－生物検定法は比の統計学であった－」の

中で「出現率の差および比の信頼区間」として取り上げてきました。 

 

  今回は、計数値の難しい解析の代表例として「繁殖試験での母体当たりの胎児の奇形

発生率の群間比較」があります。この例を用いて「計数データの解析の進化」を紹介し

てみたいと思います。JMP で解析することができるのでしょうか。奇形発生率を計量値

としてみなして解析していいのでしょうか。 

 

  今回取り上げるデータは、第 11 回のセミナー「平行性検定、実験の反復」ですでに

示されています。 

 

今回の講義と実習も、D. Collett (1991)、Modeling Binary Data をベースにする。

この本は、統計モデルを用いた豊富な薬理試験、毒性試験などの生物系の実験

データの解析事例を用いつつ、2 値データの解析の理論的な背景が丁寧に解説さ

れている。生物検定法は第 4 章 Bioassay and some applications で取り上げられて

いる。同一条件で複数のバッチに対して実験をしたような場合、あるいは繁殖

試験での同一母体で胎児の異常数を扱うような場合については、第 6 章の

Overdispersion で取り上げられている。 

注）  D.colleet(1991) の Modeling Binary Data は、反応が 2 値の生物検定法につい

ての代表的な成書です。2003 年に第 2 版が出版されていて、この 10 年で目覚しい

進歩が反映されています。 

 

  繁殖試験での同一母体で胎児の異常数を扱うような場合は、第 1 章の Example 1.4 で

示されています。 

 

Example 1.4 A toxicological study  

In an article on, the design of toxicological experiments involving reproduction, Weil (1970) 

describes a study in which female rats are fed different diets during pregnancy. One group of 

sixteen pregnant rats were fed a diet containing a certain chemical while a similar control group 

received the same diet without the addition of the chemical. After the birth of the litters, the 

number of pups which survived the twenty-one-day lactation period, expressed as a fraction of 

those alive four days after birth, was recorded; the data obtained are reproduced in Table 1.4.  

 

 
注）Treated rats と Control rats が逆？ 第 1 版でも同じ。SAS の NLMIXED のマニュ
アルのExample 46.2: Probit-Normal Model with Binomial Dataとして引用されているが、
群は逆になっているので、逆にして例と用いる。 

 

 The object of this experiment was to investigate differences between the control and treated 

rats in terms of mortality over the lactation period. In this experiment, the pregnant female rats 

were independently assigned to the two treatment groups, although a response variable is 

measured on each of the offsprings rather than on the mother. Because of genetic, environmental 

and social influences, the offsprings in a particular litter will tend to be very similar to one 

another. The binary responses from the animals of a given litter will therefore be correlated and 

unless proper account is taken of this, the precision of the estimated treatment difference is 

likely to be overestimated. 

 Experimental work in a variety of areas gives rise to problems similar to those raised by 

Example 1.4. Particular examples include the measurement of the proportion of fertile eggs 

produced by hens receiving different volumes of inseminate, binary measurements made on 

individual members of a household in a community health project, and dental surveys that 

involve counting the number of individual teeth that have caries in children using different types 

of toothpaste. 

 

Weil,C.S.(1970) Selection of the valid number of sampling units and a consideration  

  of their combination in toxicological studies involving reproduction, teratogenesis  

  or carcinogenesis. Food and Cosmetics Toxicology, 8, 177-182. 

2. JMP による解析 

2.1. データの特徴 

  Collett の table 4.1 を JMP で解析できるように入力し直し、生存胎児率について経験

ロジット変換を行い平均値の差の検定を試みてみよう。 

 

表 2.1  生存胎児数 JMP データ（Collett104.jmp） 

図 2.1  群ごとの生存胎児の割合と経験ロジット変換による作図 

 

  図 2.1 に示すように経験ロジット変換データに対して平均値の差の検定を試みるが

有意な差とはならない。JMPに用意されているノンパラメトリック検定では有意な差と

はならないが、0.60 以下の生存率となっていた母体数は、test群では 16 匹中 5 匹であり、

図 2.2 に示すように事後的な 2x2 の有意差検定を適用すると有意な差となる。念のた

めに分散の違いの検定を適用すると、図 2.3 に示すようにBartlettの検定でp=0.0289 と

なり、等分散性とは言いきりがたい。 

図 2.2  事後的な 2x2 での有意差検定の適用 

図 2.3  事後的分散の違いの検定 

 異常胎児数を延べのしてみると、図 2.4 に示すように、対照群は 16/158=0.1013、テス

ト薬群では 33/145=0.2276 と 12.6％の差があり、統計的にも明らかな差が認められる。 

図 2.4  延べ胎児数での 2x2 での有意差検定の適用 

 

  延べにした解析は、この例のように、異常な胎児が生まれる母獣からは異常胎児が生

まれやすいという関連があり、統計的に差が出やすくなることが、同腹効果として経験

的に知られている。図 2.1 のボックス・プロットから、異常が出るラットからは、多

くの異常胎児が観察されるが、対照群と同程度のラットも多数いることが観察される。 

このことから、異常胎児がテスト薬群の母獣から一定の割合で出生するのではなく、異

常胎児を出生する母獣かまとめて出生していることが観察されている。したがって、延

べにした場合には、統計的に有意になりやすい。母獣ごとの異常発生率に対する平均値

の差の比較の場合にも、この現象が反映して、逆に統計的には平均値の差が検出しにく

くなる。 

 

2.2. ロジスティック回帰 

  JMPでロジスティック回帰を試みる。表 2.1 のr/nタイプのデータ構造から、反応y（0、

1）に変換した結果を 表 2.2 に示す。ロジスティック回帰で、薬剤群を固定効果、母

獣ひと腹ごとの番号（litter）をランダム効果として解析してみよう。図 2.5 にモデルの

あてはめ、図 2.6 に結果を示す。この結果は、なにかおかしいことに気がつく。Effect 

of Wald Testの結果、p=0.9576 となる。実は、JMPでランダム効果を考慮した解析は反応

が計量値のみの場合で、2 値反応のロジスティック回帰ではメッセージなしに苦しまぎ

れな結果の出力をしているのであり、結果は正しくない。 

表 2.2 ロジスティック回帰用データセット 

 

 
 

 

図 2.5  JMP によるモデルのあてはめ 

図 2.6  JMP によるロジスティック回帰の結果 

ランダム効果モデルは正しく行われていない 

 

 

  母獣一腹を無視して薬剤群のみの 2 水準のロジスティック回帰の結果を 図 2.7 に示

す。Whole Model Testの尤度比検定のp値は 0.0027 と 図 2.4 の 2x2 表に対する尤度比

検定のp値に一致している。  

図 2.7  JMP によるロジスティック回帰の結果 

Litter をモデルに入れなければ正しくない 

2.3. 反応（0、1）を計量値とした場合の枝分かれ型の解析 

  JMPでは 2 値反応では、ランダム効果を考慮した解析ができないので、2 値反応を計

量値とみなした実験データとして便宜的に解析してみよう。表 2.3 は典型的な枝分か

れ構造を持つ実験データであり、計量的な反応とみなして逐次型の分散分析を試みる。

解析すべき生データの本質は、このような 0，1 の 2 値データである。 

 

表 2.3  分割実験とみなしたデータ構造 
trt litter 1 2 3 4 5 6 7 8 9 10 11 12 13 0の数 1の数 n
c 1 0 0 0 0 0 0 0 0 0 0 0 0 0 13 0 13
c 2 0 0 0 0 0 0 0 0 0 0 0 0 12 0 12
c 3 0 0 0 0 0 0 0 0 0 9 0 9
c
c 15 0 0 0 0 0 0 0 0 1 1 1 7 3 10
c 16 0 0 0 0 0 0 0 0 1 1 1 7 3 10
t 17 0 0 0 0 0 0 0 0 0 0 0 0 12 0 12
t 18 0 0 0 0 0 0 0 0 0 0 0 11 0 11
t 19 0 0 0 0 0 0 0 0 0 0 10 0 10
t
t 31 0 0 0 1 1 1 1 1 1 1 3 7 10
t 32 1 1 1 1 1 1 1 0 7 7  

胎児の異常を 1、正常を 0 とした 
 

 

  

図 2.8  JMP による逐次検定型の分散分析実施 

 

  この方法は、母獣ごとの出生率を計算して、2 群の t 検定を行ったと同様の手順であ

り、異常が出る母獣には異常な胎児が多く出やすいという傾向は考慮されていない。 

図 2.9  逐次型分散分析表（このままでは使えない） 

 

表 2.4  枝分かれ型分散分析 

Source DF ss ms F p
Model 31 13.7697 0.4442 4.4083 0.0000
(trt) 1 1.2065 1.2065 2.8811 0.1000
(litter) 30 12.5632 0.4188 4.1561 0.0000
Error 271 27.3062 0.1008
C. Total 302 41.0759 .  

Excel で再計算：trt は litter の ms で検定． 

3. SAS/LOGISTIC による反復のある 2 値データの解析 

  ロジスティック回帰を主体にしてきたプログラム、SAS の LOGISTIC プロシジャも

JMP と同様に、反復のある 2 値の実験データの解析を適切に解析することはできなかっ

たが、LOGISTIC プロシジャで CLASS ステートメントが使えるようになったときに、

CLASS 変数で指定された変数について、その水準ごとの分散を逆数とした重みつき最

尤法がサポートされるようになった。 

  これについては、11 回のセミナーでも過大発散 overdispersion の問題の解決策として

取り上げてきた課題であり、今回は、反復のある 2 値データの解析の基礎の問題として

取り上げる。早速、SAS による解析方法と結果の見方を学習しよう。 

 

  SAS Program 1a.   データの読み込み 

Title 'rats_litter.sas    2005-03-22    Y.Takahashi' ; 
  
data rats; 
      input trt$ litter r  n  @@ ; 
datalines; 
c 1 13 13   c 2 12 12   c 3 9 9 
c 4 9 9    c 5 8 8    c 6 8 8 
c 7 12 13   c 8 11 12   c 9 9 10 
c 10 9 10   c 11 8 9    c 12 11 13 
c 13 4 5    c 14 5 7    c 15 7 10 
c 16 7 10 
t 17 12 12   t 18 11 11   t 19 10 10 
t 20 9 9    t 21 10 11   t 22 9 10 
t 23 9 10   t 24 8 9    t 25 8 9 
t 26 4 5    t 27 7 9    t 28 4 7 
t 29 5 10   t 30 3 6    t 31 3 10 
t 32 0 7 
; 
proc print data=rats ; 
run ; 
 

 

3.1. 反復の自由度で調整する方法 

  LOGISTIC プロシジャで反復のある 2 値データの解析は、model ステートメントで左

辺を r/n 型の反応として指定することにより可能となる（aggregate オプションで一般

の反応型でも対応は可能である）が、Progrmam1b に示すように単純に主効果として trt

を指定するだけでは、littter ごとに入力されたデータであっても、解析時には延べにし

た全体についてロジスティック回帰が行われてしまう。 

  SAS Program 1b.  litter を無視した延べデータでの解析  

proc logistic data=rats ; 
    class   trt litter / param=ref ref=first ; 
 model   r / n = trt ; 
run ; 

   

  Litter 単位（クラスター）での解析のためには、model ステートメントで scale オプショ

ンを使う必要がある。scale=deviance と指定することにより反応を 0、1 とした枝分かれ

型の分散分析と同様の考えに従ったロジスティック回帰が行われる。 

 

  SAS Program 1c.  litter 効果を考慮した分散分析的解析 

proc logistic data=rats ; 
    class   trt litter / param=ref ref=first ; 
 model   r / n = trt / scale=deviance ; 
run ; 

 

このオプションによる結果、Output 1a に示すように、Deviance の調整値が

value/DF=2.8729 と計算されている。これは、32 の litter の Deviance が自由度 30 の 分

布に従うとしたときに、86.1871 となり、2 項分布に従うとはみなせないので、Control

と trt の差についての Deviance が自由度 1 の 分布に従うとするのではなく、

value/DF=2.8729 で割った値が自由度 1 の 分布に従うと考えるのである。 

2χ

2χ
2χ

 

     SAS Program 1a. 
                 Deviance and Pearson Goodness-of-Fit Statistics 
          Criterion          Value       DF     Value/DF     Pr > ChiSq 
          Deviance         86.1871       30       2.8729         <.0001 
          Pearson          80.6353       30       2.6878         <.0001 
                    Number of events/trials observations: 32 
NOTE: The covariance matrix has been multiplied by the heterogeneity factor 
      (Deviance / DF) 2.8729. 

 

Program 1b で計算した延べでのロジスティック回帰の結果は、 となり、自

由度 1 の 分布に従うとみなすと、

4946.82 =χ
2χ 0036.0=p と高度に有意となる。 

 

     SAS Program 1b. 
               Analysis of Maximum Likelihood Estimates 
                                  Standard          Wald 
 Parameter      DF    Estimate       Error    Chi-Square    Pr > ChiSq 
 Intercept       1      2.1832      0.2637       68.5415        <.0001 
 trt       t     1     -0.9612      0.3298        8.4946        0.0036 

注）(-0.9612)^2/0.3298^2 = 8.49427 

  



― 12 ―  

 

これは、litter 効果を無視した結果であり、litter 間の過大発散を割り引くために、litter

効果 value/DF=2.8729 で、trt の 値をわって、それが自由度 1 の 分布に従うとする

のである。この結果、trt の効果は、

2χ 2χ

0855.0=p となる。 

 

     SAS Output 1c.  Litter 効果を考慮した解析 
               Analysis of Maximum Likelihood Estimates 
                                  Standard          Wald 
 Parameter      DF    Estimate       Error    Chi-Square    Pr > ChiSq 
 Intercept       1      2.1832      0.4470       23.8579        <.0001 
 trt       t     1     -0.9612      0.5590        2.9568        0.0855 

 
注）8.49427/2.8729 = 2.9568 

 

SAS のマニュアルの抜粋 
SCALE= scale  

enables you to supply the value of the dispersion parameter or to specify the 
method for estimating the dispersion parameter. It also enables you to display the 
"Deviance and Pearson Goodness-of-Fit Statistics" table. To correct for 
overdispersion or underdispersion, the covariance matrix is multiplied by the 
estimate of the dispersion parameter. Valid values for scale are as follows:  

D  |  DEVIANCE  
specifies that the dispersion parameter be estimated by the deviance divided by its 
degrees of freedom.  

P  |  PEARSON  
specifies that the dispersion parameter be estimated by the Pearson chi-square 
statistic divided by its degrees of freedom.  

WILLIAMS <(constant)>  
specifies that Williams' method be used to model overdispersion. This option can 
be used only with the events/trials syntax. An optional constant can be specified as 
the scale parameter; otherwise, a scale parameter is estimated under the full model. 
A set of weights is created based on this scale parameter estimate. These weights 
can then be used in fitting subsequent models of fewer terms than the full model. 
When fitting these submodels, specify the computed scale parameter as constant. 
See Example 42.9 for an illustration.  

N  |  NONE  
specifies that no correction is needed for the dispersion parameter; that is, the 
dispersion parameter remains as 1. This specification is used for requesting the 
deviance and the Pearson chi-square statistic without adjusting for overdispersion.  

constant  

sets the estimate of the dispersion parameter to be the square of the given constant. 
For example, SCALE=2 sets the dispersion parameter to 4. The value constant 
must be a positive number.  

 
You can use the AGGREGATE (or AGGREGATE=) option to define the 
subpopulations for calculating the Pearson chi-square statistic and the deviance. In 
the absence of the AGGREGATE (or AGGREGATE=) option, each observation is 
regarded as coming from a different subpopulation. For the events/trials syntax, 
each observation consists of n Bernoulli trials, where n is the value of the trials 
variable. For single-trial syntax, each observation consists of a single response, and 
for this setting it is not appropriate to carry out the Pearson or deviance 
goodness-of-fit analysis. Thus, PROC LOGISTIC ignores specifications SCALE=P, 
SCALE=D, and SCALE=N when single-trial syntax is specified without the 
AGGREGATE (or AGGREGATE=) option.  
 
The "Deviance and Pearson Goodness-of-Fit Statistics" table includes the Pearson 
chi-square statistic, the deviance, their degrees of freedom, the ratio of each 
statistic divided by its degrees of freedom, and the corresponding p-value. For more 
information, see the "Overdispersion" section.  

 

AGGREGATE  
AGGREGATE= (variable-list)  

specifies the subpopulations on which the Pearson chi-square test statistic and the 
likelihood ratio chi-square test statistic (deviance) are calculated. Observations with 
common values in the given list of variables are regarded as coming from the same 
subpopulation. Variables in the list can be any variables in the input data set. 
Specifying the AGGREGATE option is equivalent to specifying the AGGREGATE= 
option with a variable list that includes all explanatory variables in the MODEL 
statement. The deviance and Pearson goodness-of-fit statistics are calculated only 
when the SCALE= option is specified. Thus, the AGGREGATE (or AGGREGATE=) 
option has no effect if the SCALE= option is not specified. See the section 
"Rescaling the Covariance Matrix" for more detail.  

 

 

Overdispersion 

For a correctly specified model, the Pearson chi-square statistic and the deviance, divided 
by their degrees of freedom, should be approximately equal to one. When their values are 
much larger than one, the assumption of binomial variability may not be valid and the data 
are said to exhibit overdispersion. Underdispersion, which results in the ratios being less 
than one, occurs less often in practice.  

When fitting a model, there are several problems that can cause the goodness-of-fit 
statistics to exceed their degrees of freedom. Among these are such problems as outliers in 

the data, using the wrong link function, omitting important terms from the model, and 
needing to transform some predictors. These problems should be eliminated before 
proceeding to use the following methods to correct for overdispersion.  

Rescaling the Covariance Matrix 
One way of correcting overdispersion is to multiply the covariance matrix by a dispersion 
parameter. This method assumes that the sample sizes in each subpopulation are 
approximately equal. You can supply the value of the dispersion parameter directly, or you 
can estimate the dispersion parameter based on either the Pearson chi-square statistic or 
the deviance for the fitted model.  

The Pearson chi-square statistic and the deviance are given by  

 

where m is the number of subpopulation profiles, k+1 is the number of response levels, rij is 

the total weight (sum of the product of the frequencies and the weights) associated with jth 

level responses in the ith profile, , and is the fitted probability for 

the jth level at the ith profile. Each of these chi-square statistics has mk - p degrees of 

freedom, where p is the number of parameters estimated. The dispersion parameter is 

estimated by  

 

In order for the Pearson statistic and the deviance to be distributed as chi-square, there 
must be sufficient replication within the subpopulations. When this is not true, the data are 
sparse, and the p-values for these statistics are not valid and should be ignored. Similarly, 
these statistics, divided by their degrees of freedom, cannot serve as indicators of 
overdispersion. A large difference between the Pearson statistic and the deviance provides 
some evidence that the data are too sparse to use either statistic.  
You can use the AGGREGATE (or AGGREGATE=) option to define the subpopulation 
profiles. If you do not specify this option, each observation is regarded as coming from a 
separate subpopulation. For events/trials syntax, each observation represents n Bernoulli 
trials, where n is the value of the trials variable; for single-trial syntax, each observation 
represents a single trial. Without the AGGREGATE (or AGGREGATE=) option, the Pearson 
chi-square statistic and the deviance are calculated only for events/trials syntax.  
Note that the parameter estimates are not changed by this method. However, their standard 
errors are adjusted for overdispersion, affecting their significance tests.  
Williams' Method 
Suppose that the data consist of n binomial observations. For the ith observation, let ri/ni be 

the observed proportion and let xi be the associated vector of explanatory variables. 
Suppose that the response probability for the ith observation is a random variable Pi with 
mean and variance  

 

where pi is the probability of the event, and is a nonnegative but otherwise unknown 
scale parameter. Then the mean and variance of ri are  

 

Williams (1982) estimates the unknown parameter by equating the value of Pearson's 
chi-square statistic for the full model to its approximate expected value. Suppose wi* is the 
weight associated with the ith observation. The Pearson chi-square statistic is given by  

 

Let g'(·) be the first derivative of the link function g(·). The approximate expected value of 

is  

 

where and di is the variance of the linear predictor 

. The scale parameter is estimated by the following iterative procedure.  

At the start, let wi*=1 and let be approximated by ri/ni, i = 1,2, ... ,n. If you apply these 

weights and approximated probabilities to and and then equate them, an initial 

estimate of is therefore  

 

where p is the total number of parameters. The initial estimates of the weights become 

. After a weighted fit of the model, is recalculated, and so is 

. Then a revised estimate of is given by  

 

The iterative procedure is repeated until is very close to its degrees of freedom.  

Once has been estimated by under the full model, weights of can 
be used in fitting models that have fewer terms than the full model. See Example 42.9 for 

an illustration.  

3.2. 重みつきロジスティック回帰 

  反復（クラスター）がある 2 値テータの場合にその自由度で調整する方法は、それぞ

れの反復内での観測データ数が同じであることを前提にしている。実際には、生まれる

子の数は不均一であり、同じ litter の子の数（litter サイズ）が大きいときに重みを相対

的に小さくするような調整法が Williams(1982)によって定式化され、SAS の LOGISTIC

プロシジャでは、model ステートメントで scale=williams オプションを設定することに

より計算できる。Williams の方法は、それぞれの litter の分散を、通常の 2 項誤差に負

でない未知のスケールパラメータφ を用いて 

 

とし、 の期待値と分散を ir

 

と考え、φ をデーから推定し、重みを 

 

とする解析方法である。 

 

  SAS Program 2.  litter 効果を考慮した Williams の調整 

proc logistic data=rats ; 
    class   trt litter / param=ref ref=first ; 
 model   r / n = trt / scale=williams ; 
run ; 

 

  SAS Output 2a.  Williams の重みの推定 

                             The LOGISTIC Procedure 
 
                                Model Information 
          Data Set                       WORK.RATS 
          Response Variable (Events)     r 
          Response Variable (Trials)     n 
          Weight Variable                1 / ( 1 + 0.202805 * (n - 1) ) 
          Model                          binary logit 
          Optimization Technique         Fisher's scoring 
 

 

  推定されたスケールパラメータφ は、Output 2aから、0.2028 である。実際のデータに

ついて計算してみよう。図 3.1 に示すように、litterサイズが実際に 5 から 13 と広がり

があり、重みは 0.55 から 0.29 の範囲となる。 

図 3.1  William の重み 

 

  LOGISTIC プロシジャは、スケールパラメータの推定に引き続き重みつきのロジス

ティッ回帰を行う。結果を Output 2b に示す。推定値は、-1.0205、p 値は 0.0581 となる。 

 

  SAS Output 2b.  Williams の重み付ロジスティック回帰の結果 

                    Analysis of Maximum Likelihood Estimates 
                                 Standard          Wald 
Parameter      DF    Estimate       Error    Chi-Square    Pr > ChiSq 
Intercept       1      2.1439      0.4370       24.0700        <.0001 
trt       t     1     -1.0205      0.5386        3.5897        0.0581 

 

  重みを data ステップで計算し、同じ結果が重みつきロジスティック回帰で得られる

か検証してみよう。 Program 3 に示すようにそれぞれの litter ごとに重み w を計算し、

weight ステートメントで w を指定する。結果は、スケールパラメータで williams を指定

したときと同じ結果が得られることが確認できる。 

 

SAS Program 3.  重みの計算 
data rats2 ; 
    set rats ; 
 w = 1 / ( 1 + 0.202805 * (n - 1) ) ; 
 
proc print data=rats2 ; 
run ; 
 
proc logistic data=rats3 ; 
    class  trt / param=ref ref=first ; 
 weight w ; 
 model  r / n = trt  ; 
run ; 

SAS Output 3.  重みつきロジスティック回帰 
OBS    trt    litter     r     n       w 
 
  1     c        1      13    13    0.29123 
  2     c        2      12    12    0.30952 
  3     c        3       9     9    0.38132 
  4     c        4       9     9    0.38132 
  5     c        5       8     8    0.41329 
  6     c        6       8     8    0.41329 
  7     c        7      12    13    0.29123 
  8     c        8      11    12    0.30952 
  9     c        9       9    10    0.35395 
 10     c       10       9    10    0.35395 
 11     c       11       8     9    0.38132 
 12     c       12      11    13    0.29123 
 13     c       13       4     5    0.55211 
 14     c       14       5     7    0.45109 
 15     c       15       7    10    0.35395 
 16     c       16       7    10    0.35395 
 17     t       17      12    12    0.30952 
 18     t       18      11    11    0.33025 
 19     t       19      10    10    0.35395 
 20     t       20       9     9    0.38132 
 21     t       21      10    11    0.33025 
 22     t       22       9    10    0.35395 
 23     t       23       9    10    0.35395 
 24     t       24       8     9    0.38132 
 25     t       25       8     9    0.38132 
 26     t       26       4     5    0.55211 
 27     t       27       7     9    0.38132 
 28     t       28       4     7    0.45109 
 29     t       29       5    10    0.35395 
 30     t       30       3     6    0.49652 
 31     t       31       3    10    0.35395 
 32     t       32       0     7    0.45109 

 
 
               Analysis of Maximum Likelihood Estimates 
                                  Standard          Wald 
 Parameter      DF    Estimate       Error    Chi-Square    Pr > ChiSq 
 Intercept       1      2.1439      0.4370       24.0700        <.0001 
 trt       t     1     -1.0205      0.5386        3.5897        0.0581 

4. 変量効果モデル 

  SAS の例題で、litter 効果のデータが使われている。例示として、ロジット変換では

なく、プロビット変換となっているが、検証のためにロジット変換はしていない。 

 

 

4.1. SAS の NLMIXED を用いた線形ロジスティックモデルでのランダム効果 

  我々は，過大発散が，同一に処理された実験ユニットの異なるバッチについての反応

確率の変動で説明できることを見てきた．6.1 節（Collett: 1st ed.）で説明したように，こ

の変動は，記録されていない関連する説明変数に起因すると考えることができるか，あ

るいは，適切に測定されるか制御されないような確かな説明変数に含まれる． 

 

SAS/NLMEXED  Example 46.2: Probit-Normal Model with Binomial Data 

For this example, consider the data from Weil (1970), also studied by Williams (1975), Ochi and 

Prentice (1984), and McCulloch (1994). In this experiment 16 pregnant rats receive a control 

diet and 16 receive a chemically treated diet, and the litter size for each rat is recorded after 4 

and 21 days. The SAS data set is a follows.  

 
   data rats; 
      input trt$  m  x @@ ; 
      if (trt='c') then do; 
         x1 = 1; 
         x2 = 0; 
      end; 
      else do; 
         x1 = 0; 
         x2 = 1; 
      end; 
      litter = _n_; 
      datalines; 
   c 13 13   c 12 12   c  9  9   c  9  9   c  8  8   c  8  8 
   c 13 12   c 12 11   c 10  9   c 10  9   c  9  8   c 13 11 
   c  5  4   c  7  5   c 10  7   c 10  7 
   t 12 12   t 11 11   t 10 10   t  9  9   t 11 10   t 10  9 
   t 10  9   t  9  8   t  9  8   t  5  4   t  9  7   t  7  4 
   t 10  5   t  6  3   t 10  3   t  7  0 
   run; 

図 4.1  リッターサイズ m と生存胎児数の確率楕円 

図 4.2 生存胎児数の割合と経験ロジットでの平均値の比較 

 

Here, M represents the size of the litter after 4 days, and X represents the size of the litter after 

21 days. Also, indicator variables X1 and X2 are constructed for the two treatment levels.  

Following McCulloch (1994), assume a latent survival model of the form  

 
where i indexes treatment, j indexes litter, and k indexes newborn rats within a litter. The ti 

represent treatment means, the represent random litter effects assumed to be iid N(0,s )i
2 , and 

the eijk represent iid residual errors, all on the latent scale.  

Instead of observing the survival times yijk, assume that only the binary variable indicating 

whether yijk exceeds 0 is observed. If xij denotes the sum of these binary variables for the ith 

treatment and the jth litter, then the preceding assumptions lead to the following generalized 

linear mixed model:  

where mij is the size of each litter after 4 days and  

 

The PROC NLMIXED statements to fit this model are as follows.  
 
   proc nlmixed data=rats; 
      parms t1=1 t2=1 s1=.05 s2=1; 
      eta = x1*t1 + x2*t2 + alpha; 
      p   = probnorm(eta); 
      model      x ~ binomial(m,p); 
      random    alpha ~ normal(0, x1*s1*s1+x2*s2*s2) subject=litter; 
      estimate 'gamma2' t2/sqrt(1+s2*s2); 
      predict   p out=p; 
   run; 
 

As in the previous example, the PROC NLMIXED statement invokes the procedure and the 

PARMS statement defines the parameters. The parameters for this example are the two 

treatment means, T1 and T2, and the two random-effect standard deviations, S1 and S2.  

The indicator variables X1 and X2 are used in the program to assign the proper mean to each 

observation in the input data set as well as the proper variance to the random effects. Note that 

programming expressions are permitted inside the distributional specifications, as illustrated by 

the random-effects variance specified here.  

The ESTIMATE statement requests an estimate of , which is a location-scale 

parameter from Ochi and Prentice (1984).  

The PREDICT statement constructs predictions for each observation in the input data set. For 

this example, predictions of P and approximate standard errors of prediction are output to a SAS 

data set named P. These predictions are functions of the parameter estimates and the empirical 

Bayes estimates of the random effects .  

The output for this model is as follows.  
 
                           Specifications 
  Data Set                                    DD.RATS 
  Dependent Variable                          x 
  Distribution for Dependent Variable         Binomial 
  Random Effects                              alpha 

  Distribution for Random Effects             Normal 
  Subject Variable                            litter 
  Optimization Technique                      Dual Quasi-Newton 
  Integration Method                          Adaptive Gaussian 
                                              Quadrature 
The "Specifications" table provides basic information about this nonlinear mixed model.  
 
 
                             Dimensions 
              Observations Used                     32 
              Observations Not Used                  0 
              Total Observations                    32 
              Subjects                              32 
              Max Obs Per Subject                    1 
              Parameters                             4 
              Quadrature Points                      7 
 
The "Dimensions" table provides counts of various variables.  
 
 
                             Parameters 
           t1          t2          s1          s2    NegLogLike 
            1           1        0.05           1    54.9362323 
 
The "Parameters" table lists the starting point of the optimization.  
 
 
                          Iteration History 
  Iter     Calls    NegLogLike        Diff     MaxGrad       Slope 
     1         2    53.9933934    0.942839    11.03261    -81.9428 
     2         3     52.875353     1.11804    2.148952    -2.86277 
     3         5    52.6350386    0.240314    0.329957    -1.05049 
     4         6    52.6319939    0.003045    0.122926    -0.00672 
     5         8    52.6313583    0.000636    0.028246    -0.00352 
     6        11    52.6313174    0.000041    0.013551    -0.00023 
     7        13    52.6313115    5.839E-6    0.000603    -0.00001 
     8        15    52.6313115     9.45E-9    0.000022    -1.68E-8 
            NOTE: GCONV convergence criterion satisfied. 
The "Iterations" table indicates successful convergence in 8 iterations.  
 
                           Fit Statistics 
              -2 Log Likelihood                  105.3 
              AIC (smaller is better)            113.3 
              AICC (smaller is better)           114.7 
              BIC (smaller is better)            119.1 
 
The "Fitting Information" table lists some useful statistics based on the maximized value 
of the log likelihood.  
 
                                    Parameter Estimates 
                     Standard 
Parameter  Estimate     Error    DF  t Value  Pr > |t|   Alpha     Lower     Upper  Gradient 
 
t1           1.3063    0.1685    31     7.75    <.0001    0.05    0.9626    1.6499  -0.00002 
t2           0.9475    0.3055    31     3.10    0.0041    0.05    0.3244    1.5705  9.283E-6 
s1           0.2403    0.3015    31     0.80    0.4315    0.05   -0.3746    0.8552  0.000014 
s2           1.0292    0.2988    31     3.44    0.0017    0.05    0.4198    1.6385  -3.16E-6 
 
The "Parameter Estimates" table indicates significance of all of the parameters except 
S1.  
 
 
 
 
 



  ― 23 ― 

                                    Additional Estimates 
 
                       Standard 
  Label     Estimate      Error     DF   t Value   Pr > |t|    Alpha      Lower      Upper 
  gamma2      0.6603     0.2165     31      3.05     0.0047     0.05     0.2186     1.1019 
  t2 - t1    -0.3588     0.3489     31     -1.03     0.3117     0.05    -1.0703     0.3528 

 

The "Additional Estimates" table displays results from the ESTIMATE statement. The estimate 

of equals 0.6602, agreeing with that obtained by McCulloch (1994). The standard error 
0.2166 is computed using the delta method (Billingsley 1986).  

  Not shown is the P data set, which contains the original 32 observations and predictions of the 

pij.  

 

表 4.1  ランダム係数の推定 
コントロール 投与群 

trt litter m x x1 x2 alpha^ eta^ p^ trt litter m x x1 x2 alpha^ eta^ p^ 
c 1 13 13 1 0 0.1177 1.424 0.9228 t 17 12 12 0 1 0.9182 1.8657 0.969 
c 2 12 12 1 0 0.11 1.4163 0.9217 t 18 11 11 0 1 0.8895 1.837 0.9669
c 3 9 9 1 0 0.0857 1.392 0.918 t 19 10 10 0 1 0.8582 1.8057 0.9645
c 4 9 9 1 0 0.0857 1.392 0.918 t 20 9 9 0 1 0.8238 1.7713 0.9617
c 5 8 8 1 0 0.0772 1.3835 0.9167 t 21 11 10 0 1 0.3084 1.2559 0.8954
c 6 8 8 1 0 0.0772 1.3835 0.9167 t 22 10 9 0 1 0.2633 1.2108 0.887 
c 7 13 12 1 0 0.0223 1.3286 0.908 t 23 10 9 0 1 0.2633 1.2108 0.887 
c 8 12 11 1 0 0.0138 1.3201 0.9066 t 24 9 8 0 1 0.2129 1.1604 0.8771
c 9 10 9 1 0 -0.0041 1.3022 0.9036 t 25 9 8 0 1 0.2129 1.1604 0.8771
c 10 10 9 1 0 -0.0041 1.3022 0.9036 t 26 5 4 0 1 -0.0763 0.8712 0.8082
c 11 9 8 1 0 -0.0133 1.293 0.902 t 27 9 7 0 1 -0.1516 0.7959 0.787 
c 12 13 11 1 0 -0.0669 1.2394 0.8924 t 28 7 4 0 1 -0.6313 0.3162 0.6241
c 13 5 4 1 0 -0.053 1.2533 0.8949 t 29 10 5 0 1 -0.8251 0.1224 0.5487
c 14 7 5 1 0 -0.1276 1.1787 0.8807 t 30 6 3 0 1 -0.7596 0.1879 0.5745
c 15 10 7 1 0 -0.1828 1.1235 0.8694 t 31 10 3 0 1 -1.2687 -0.3212 0.374 
c 16 10 7 1 0 -0.1828 1.1235 0.8694 t 32 7 0 0 1 -1.9976 -1.0501 0.1468

Parameter({t1 = 1.3063, t2 = 0.9475},  :x1 * t1 +  :x2 * t2 +  :Name("alpha^")) 

Normal Distribution( :Name("eta^")) 

 

  



― 24 ―  

e
ta

^
-1.5

-1.0

-0.5

0.0

0.5

1.0

1.5

2.0

c t

trt
 

図 4.3  ランダム効果を調整した推定値 ˆiη  

図 4.4  リッターサイズを含めたランダム効果を調整した胎児の生存率 

■：生データによる生存率，○：ランダム効果を調整した胎児の生存率 

5. 手法の比較（Collett） 

  Litter（同腹）効果の調整法として、自由度による調整、Williams の重みの使用、ラン

ダム効果モデルの使用を試みてきたのであるが、自由度による調整は、litter サイズが異

なる場合には問題があり、使えない。ランダム効果モデルには魅力を感ずるものの、使

いかが煩雑であるにもかかわらず、litter のランダム効果として、プロビット変換あるい

は ロ ジ ッ ト 変 換 値 に つ い て ラ ン ダ ム 効 果 が 正 規 分 布 alpha ~ normal(0, 

x1*s1*s1+x2*s2*s2) となることを仮定している。従って、Litter 効果のような場合には、

William の 

 

 

 

  どちらが一般的に望ましいのか、Colette の著書に手法の比較があるので、見てみよう。 

 

 

  この章では，我々は，過大発散のモデルに使うことができる多くの手法を検討してき

た．便宜的に，次に一覧されている． 

 

 手法 1： ウィリアム法（6.4 節） 

 手法 2： 不均一要素の使用（自由度による調整）（6.5 節） 

 手法 3： ベータ 2 項モデル（6.6 節） 

 手法 4： ランダム効果（6.7 節） 

 手法 5： モデル 4 の近似（6.7.2 節） 

 

  我々は，手法 3 が，手法 1 に基づく形の手順に比べて柔軟性に乏しいこと見てきた．

そして，この手法 3 の手順は，さらに検討される必要はない．手法 5 は，しばしば手法

1 に非常に似た結果を与え，近似が，ランダム効果の分散が小さいときにのみ妥当であ

り，手法 5 は，一般的に勧めることができない．なおまた，ランダム効果を含むモデル

のあてはめで使用できるコンピュータ ソフトウェアは，より広く利用可能となってき

ており，もはや正確なアプローチより簡単に実践できるような近似的な方法にたよるこ

とはない．類似した議論は，手法 2 で適用することができ，それは，さらにまた簡単に

適用できるのであるが，2 項分母 （the binomial denominators）がすべて等しいときにの

み妥当である． 

  手法 4 は，関係ある説明変数の値が記録されていないときや，モデルの中の変数が適

切に測定されていないかコントロールされていないときに， 過大発散のモデル化の自

然な方法である．これは，また，2 値応反応データでの過大発散のモデルに用いること

が考慮されているただ一つの手法である．あいにく，ランダム効果を含むモデルのあて

はめた後では，過大発散が実際に適用されているかどうかは明快ではない．手法 1 は，

反応確率間の変動があるときのグループ化された 2 値データの過大発散のモデリング

の一般的な手順であり，手法 2 の特殊な場合を含んでいる．手法 1 と手法 4 は，それゆ

えに一般的に使用として勧められる手順である． 

 

 

  ランダム効果モデルによる litter 効果の調整法は、各群内で、平均的な反応よりも大

きいか小さいかによって調整がなされていて、litter サイズの大きさは配慮されていない。 

図 5.1  ランダム効果による調整 

各 litter の反応の平均からのずれの大きさに比例した推定値の調整、d =  pp −ˆ

図 5.2 Litter サイズに対する調整量 
各 litter サイズについて負の相関があるように見受けられるが、litter サイズ
が小さいときには反応の平均からのずれが大きいとの関連と交絡している。 

6. SAS/GENMOD プロシジャでの解析 

  SAS/ GENMOD プロシジャでも反復のある 2 値データの解析を行うことができる。ス

ケールパラメータを deviance とした解析結果を示す。LIGISTIC プロシジャの結果と一

致していることが確かめられる。 

 
proc genmod data=rats ; 
    class  trt / param=ref ref=first ; 
    model  r / n = trt  / dist=binomial link=logit scale=deviance ; 
 run ; 

 
                              The GENMOD Procedure 
 
                                Model Information 
                     Data Set                      WORK.RATS 
                     Distribution                   Binomial 
                     Link Function                     Logit 
                     Response Variable (Events)            r 
                     Response Variable (Trials)            n 
 
                     Number of Observations Read          32 
                     Number of Observations Used          32 
                     Number of Events                    254 
                     Number of Trials                    303 
 
 
                             Class Level Information 
                                                Design 
                          Class     Value     Variables 
                          trt       c                 0 
                                    t                 1 
 
                     Criteria For Assessing Goodness Of Fit 
          Criterion                 DF           Value        Value/DF 
          Deviance                  30         86.1871          2.8729 
          Scaled Deviance           30         30.0000          1.0000 
          Pearson Chi-Square        30         80.6353          2.6878 
          Scaled Pearson X2         30         28.0675          0.9356 
          Log Likelihood                      -45.1010 
   Algorithm converged. 
 
                         Analysis Of Parameter Estimates 
                              Standard       Wald 95%          Chi- 
  Parameter     DF  Estimate     Error   Confidence Limits   Square  Pr > ChiSq 
  Intercept      1    2.1832    0.4470    1.3072    3.0593    23.86      <.0001 
  trt        t   1   -0.9612    0.5590   -2.0569    0.1344     2.96      0.0855 
  Scale          0    1.6950    0.0000    1.6950    1.6950 
 
NOTE: The scale parameter was estimated by the square root of DEVIANCE/DOF. 

7. メタアナリシスへの応用 

  15 回のセミナーで 

 

2.3. ロジスティク回帰分析によるオッズ比の併合 

2.4. オッズ比の併合 

2.5. EBM メタ・アナリシスでのオッズ比 

2.6. コクラン・ライブラリ 

コクラン ライブラリの中身 

2.7. 丹後のメタ・アナリシスの紹介 

2.8. Peto の方法 

2.9. 漸近分散法－オッズ比 

2.10. Mantel-Haenszel のオッズ比 

2.11. DerSimonian-Laird－の方法－オッズ比 

2.12. ロジスティック回帰によるオッズ比の併合 

PROC GENMOD 母数モデル 

PORC LOGISTIC  母数モデル 

JMP によるロジスティック回帰 

 

を取り上げ、今回は、ランダム効果モデルと従来の方法との対比をしたかったのである

が、次回以後にまわしたい。 

